Magma and the LMFDB

Lecture 1: Lattices & Quadratic Forms

Eran Assaf

CTNT Summer School 2026

MIT

Lecture 1: Outline

Introduction to Magma

Quadratic Forms

Lattices and Theta Functions

The LMFDB

Introduction to Magma

What is Magma?

- Computer algebra system specialized in algebra, number theory, algebraic geometry, and algebraic combinatorics
- Developed at the University of Sydney
- Powerful for:
    - Number theory and algebraic number theory
    - Lattices and codes
    - Modular forms
    - Curves and abelian varieties
    - Group theory
- **Online version**: *`https://magma.maths.usyd.edu.au/calc/`*
- **Local installation**: free for non-commercial use at US institutions (via the Simons Foundation grant)

Basic Syntax

- Case-sensitive
- Statements end with a semicolon `;`
- Comments start with `//`
- Assignment: `x := value;`
- Functions: `FunctionName(arg1, arg2, ...)`

Example:

```
> // Define an integer
> n := 42;
> // A sequence
> seq := [1, 2, 3, 4, 5];
> seq[2];      // sequences are 1-indexed
2
```

Basic Types in Magma

- **Integers**: `n := 123;`
- **Rationals**: `r := 22/7;`
- **Sequences** (homogeneous): `[1, 2, 3]`
- **Lists** (heterogeneous): `[* 123, 22/7, "hello" *]`
- **Strings**: `"hello"`
- **Rings/Fields**: `Integers()`, `Rationals()`, `GF(p)`, ...

Indexing: sequences are 1-indexed.

```
> seq := [10, 20, 30];
> seq[1];      // returns 10, not 20
10
```

Tip: For help, use Online handbook, or type the function name. Use `ListSignatures` when there are many options.

Quadratic Forms

What is a Quadratic Form?

Concretely, a **quadratic form** over a field F is a homogeneous polynomial of degree 2:

$$Q(x_1, \ldots, x_r) = \sum_{i \leq j} a_{ij} x_i x_j \in F[x_1, \ldots, x_r].$$

Example: $Q(x_1, \ldots, x_r) = x_1^2 + x_2^2 + \cdots + x_r^2$.

What is a Quadratic Form?

Concretely, a quadratic form over a field F is a homogeneous polynomial of degree 2:

$$Q(x_1, \dots, x_r) = \sum_{i \leq j} a_{ij} x_i x_j \in F[x_1, \dots, x_r].$$

Example: $Q(x_1, \dots, x_r) = x_1^2 + x_2^2 + \cdots + x_r^2$.

Basis-free version (works in any characteristic, including 2):

Definition

Let V be an F-vector space. A quadratic form is a map $Q : V \to F$ such that

1. $Q(ax) = a^2 Q(x)$ for all $a \in F$, $x \in V$;
2. $B_Q(x, y) := Q(x + y) - Q(x) - Q(y)$ is bilinear.

Quadratic Forms: Key Invariants

Fix a basis $v_1, \ldots, v_r$ of V and write $G = (B_Q(v_i, v_j))_{i,j} \in M_r(F)$.

- **Gram matrix**: G satisfies $x^t G y = B_Q(x, y)$.
- **Isometry**: an isomorphism $f : V_1 \to V_2$ with $Q_2(f(x)) = Q_1(x)$. Write $Q_1 \simeq Q_2$.
- **Discriminant**: $\operatorname{disc}(Q) := \det(G) \in F/F^{\times 2}$, independent of the basis and invariant under isometry.

Quadratic Forms: Key Invariants

Fix a basis $v_1, \ldots, v_r$ of V and write $G = (B_Q(v_i, v_j))_{i,j} \in M_r(F)$.

- **Gram matrix**: G satisfies $x^t G y = B_Q(x, y)$.
- **Isometry**: an isomorphism $f : V_1 \to V_2$ with $Q_2(f(x)) = Q_1(x)$. Write $Q_1 \simeq Q_2$.
- **Discriminant**: $\operatorname{disc}(Q) := \det(G) \in F/F^{\times 2}$, independent of the basis and invariant under isometry.

Example. For $Q = x^2 + xy + y^2$,

$$G = \begin{pmatrix} 2 & 1 \\ 1 & 2 \end{pmatrix}, \qquad \operatorname{disc}(Q) = 3.$$

Warning. G is the matrix of the *polar form* B_Q, hence the 2's on the diagonal: $B_Q(v, v) = 2Q(v)$. Magma's `GramMatrix` follows this convention.

Quadratic Forms in Magma

```
> // Define a quadratic form
> R<x,y> := PolynomialRing(Rationals(), 2);
> Q := x^2 + x*y + y^2;
> V := QuadraticSpace(Q);
> // Gram matrix (note: encodes the bilinear form B_Q)
> G := GramMatrix(V);
> G;
[2 1]
[1 2]
> // Discriminant
> Determinant(G);
3
```

Equivalence of Forms

Example.

$$x^2 + y^2 + yz + 3z^2 \;\simeq\; 4x^2 + xy + y^2 + yz + 3z^2 + 6zx$$

via the substitution $(x, y, z) \mapsto (x, y, z + x)$.

Equivalence of Forms

Example.

$$x^2 + y^2 + yz + 3z^2 \;\simeq\; 4x^2 + xy + y^2 + yz + 3z^2 + 6zx$$

via the substitution $(x, y, z) \mapsto (x, y, z + x)$.

Sometimes it is easy to see that two forms are *not* equivalent:

$$x^2 + y^2 + z^2 \;\not\simeq\; x^2 - y^2 - z^2 \quad \text{over } \mathbb{R}, \quad x^2 + 5y^2 \;\not\simeq\; 2x^2 - 2xy + 3y^2 \quad \text{over } \mathbb{Q}_5.$$

(These relate to class groups, to be discussed in Lecture 2)

Equivalence of Forms

Example.

$$x^2 + y^2 + yz + 3z^2 \;\simeq\; 4x^2 + xy + y^2 + yz + 3z^2 + 6zx$$

via the substitution $(x, y, z) \mapsto (x, y, z + x)$.

Sometimes it is easy to see that two forms are *not* equivalent:

$$x^2 + y^2 + z^2 \;\not\simeq\; x^2 - y^2 - z^2 \quad \text{over } \mathbb{R}, \quad x^2 + 5y^2 \;\not\simeq\; 2x^2 - 2xy + 3y^2 \quad \text{over } \mathbb{Q}_5.$$

(These relate to class groups, to be discussed in Lecture 2)

Theorem (Minkowski, Hasse)

Quadratic forms Q_1, Q_2 over $\mathbb{Q}$ are equivalent over $\mathbb{Q}$ if and only if they are equivalent over $\mathbb{Q}_p$ for every prime p and over $\mathbb{R}$.

Example.

$$x^2 + y^2 + yz + 3z^2 \simeq 4x^2 + xy + y^2 + yz + 3z^2 + 6zx$$

via the substitution $(x, y, z) \mapsto (x, y, z + x)$.

Sometimes it is easy to see that two forms are *not* equivalent:

$$x^2 + y^2 + z^2 \not\simeq x^2 - y^2 - z^2 \quad \text{over } \mathbb{R}, \quad x^2 + 5y^2 \not\simeq 2x^2 - 2xy + 3y^2 \quad \text{over } \mathbb{Q}_5.$$

(These relate to class groups, to be discussed in Lecture 2)

Theorem (Minkowski, Hasse)

Quadratic forms Q_1, Q_2 over $\mathbb{Q}$ are equivalent over $\mathbb{Q}$ if and only if they are equivalent over $\mathbb{Q}_p$ for every prime p and over $\mathbb{R}$.

Over $\mathbb{Q}_p$, a form is classified by its rank r, discriminant $\operatorname{disc}(Q) \in \mathbb{Q}_p^\times/\mathbb{Q}_p^{\times 2}$, and the Hasse–Minkowski invariant $c_p(Q) \in \{\pm 1\}$.

Equivalence of Forms in Magma

```
> R<x,y> := PolynomialRing(Rationals(), 2);
> Q1 := x^2 + 5*y^2;
> Q2 := 2*x^2 - 2*x*y + 3*y^2;
> V1 := QuadraticSpace(Q1); V2 := QuadraticSpace(Q2);
> // Same discriminant
> Determinant(GramMatrix(V1)), Determinant(GramMatrix(V2));
20 20
> // ... but not isometric
> IsIsometric(V1, V2);
false
> // The local obstruction is at p = 5
> HasseMinkowskiInvariant(Q1, 5), HasseMinkowskiInvariant(Q2, 5);
1 -1
> // At p = 2 they agree
> HasseMinkowskiInvariant(Q1, 2), HasseMinkowskiInvariant(Q2, 2);
1 1
```

Lattices and Theta Functions

From Forms over $\mathbb{Q}$ to Lattices over $\mathbb{Z}$

Hasse–Minkowski classifies forms over $\mathbb{Q}$ locally. What happens when we tighten the picture from $\mathbb{Q}$ to $\mathbb{Z}$?

Plan. Replace the $\mathbb{Q}$-vector space (V, Q) by a $\mathbb{Z}$-submodule $\Lambda \subset V$ – a *lattice*. Suddenly there are subtler invariants, and the local-to-global principle *fails*.

What is a Lattice?

R a PID (e.g. $R = \mathbb{Z}$ or $\mathbb{Z}_p$) with fraction field F, (V, Q) a quadratic space over F.

Definition

An **R-lattice** $\Lambda \subseteq V$ is a finitely generated R-submodule containing a basis of V:

$$\Lambda = Rv_1 \oplus \cdots \oplus Rv_r,$$

where $v_1, \ldots, v_r$ are linearly independent over F.

What is a Lattice?

R a PID (e.g. $R = \mathbb{Z}$ or $\mathbb{Z}_p$) with fraction field F, (V, Q) a quadratic space over F.

Definition

An R-**lattice** $\Lambda \subseteq V$ is a finitely generated R-submodule containing a basis of V:

$$\Lambda = Rv_1 \oplus \cdots \oplus Rv_r,$$

where $v_1, \ldots, v_r$ are linearly independent over F.

- Λ is **integral** if $B_Q(x, y) \in R$ for all $x, y \in \Lambda$;
- Λ is **even** if additionally $Q(x) \in R$ for all $x \in \Lambda$.

What is a Lattice?

R a PID (e.g. $R = \mathbb{Z}$ or $\mathbb{Z}_p$) with fraction field F, (V, Q) a quadratic space over F.

Definition

An **R-lattice** $\Lambda \subseteq V$ is a finitely generated R-submodule containing a basis of V:

$$\Lambda = Rv_1 \oplus \cdots \oplus Rv_r,$$

where $v_1, \ldots, v_r$ are linearly independent over F.

- Λ is **integral** if $B_Q(x, y) \in R$ for all $x, y \in \Lambda$;
- Λ is **even** if additionally $Q(x) \in R$ for all $x \in \Lambda$.

The **Gram matrix** $G_{ij} = B_Q(v_i, v_j)$ now has entries in R. Since basis changes are by $P \in GL_r(R)$ with $\det P \in R^\times$, the **discriminant** $\mathrm{disc}(\Lambda) = \det G$ is well-defined in $R/(R^\times)^2$ – a genuine element of $\mathbb{Z}$ when $R = \mathbb{Z}$.

In Lecture 2 we will meet $\mathbb{Z}_K$-lattices: $\mathbb{Z}_K$ is a Dedekind domain, so lattices are projective (not free), and the discriminant becomes an ideal.

Examples of Lattices

Take $R = \mathbb{Z}$ and $Q(x_1, \ldots, x_n) = x_1^2 + \cdots + x_n^2$.

- **Root lattice** $D_n := \{x \in \mathbb{Z}^n : \sum x_i \in 2\mathbb{Z}\}$: even, with $\mathrm{disc}(D_n) = 4$.

Examples of Lattices

Take $R = \mathbb{Z}$ and $Q(x_1, \ldots, x_n) = x_1^2 + \cdots + x_n^2$.

- **Root lattice** $D_n := \{x \in \mathbb{Z}^n : \sum x_i \in 2\mathbb{Z}\}$: even, with $\mathrm{disc}(D_n) = 4$.
- $E_n := D_n + \mathbb{Z}(\frac{1}{2}, \ldots, \frac{1}{2})$:
  - integral iff $4 \mid n$,
  - even iff $8 \mid n$,
  - $\mathrm{disc}(E_8) = 1$ (the famous E_8 **lattice**).

Examples of Lattices

Take $R = \mathbb{Z}$ and $Q(x_1, \ldots, x_n) = x_1^2 + \cdots + x_n^2$.

- **Root lattice** $D_n := \{x \in \mathbb{Z}^n : \sum x_i \in 2\mathbb{Z}\}$: even, with $\operatorname{disc}(D_n) = 4$.
- $E_n := D_n + \mathbb{Z}(\frac{1}{2}, \ldots, \frac{1}{2})$:
  - integral iff $4 \mid n$,
  - even iff $8 \mid n$,
  - $\operatorname{disc}(E_8) = 1$ (the famous E_8 **lattice**).

Research aside. The E_8 lattice gives the densest sphere packing in dimension 8 (Viazovska, 2016; Fields Medal 2022). In dimension 24, this role is played by the **Leech lattice** – one of the 24 Niemeier lattices, and the symmetry group behind the Monster.

LMFDB: *https://alpha.lmfdb.org/Lattice/?name=E8*

Lattices in Magma

```
> // Define a lattice via basis vectors (a copy of D_3)
> basis := [[1, -1, 0], [0, 1, -1], [0, 1, 1]];
> L := Lattice(Matrix(basis));
> GramMatrix(L);
[ 2 -1 -1]
[-1  2  0]
[-1  0  2]
> Determinant(L);
4
> // Built-in root lattices
> E8 := Lattice("E", 8);
> #ShortestVectors(E8);
120
> // Note: ShortestVectors returns one of {v, -v} for each pair,
> // so the actual count of shortest vectors in E_8 is 240.
```

Isometry of Lattices

Lattices Λ_1, Λ_2 are **isometric**, written $\Lambda_1 \simeq \Lambda_2$, if there is an isometry $f : V_1 \to V_2$ with $f(\Lambda_1) = \Lambda_2$.

Isometry of Lattices

Lattices Λ_1, Λ_2 are **isometric**, written $\Lambda_1 \simeq \Lambda_2$, if there is an isometry $f : V_1 \to V_2$ with $f(\Lambda_1) = \Lambda_2$.

Local-global for lattices. If $\Lambda_1 \simeq \Lambda_2$ as $\mathbb{Z}$-lattices, then

$$\Lambda_1 \otimes \mathbb{Z}_p \simeq \Lambda_2 \otimes \mathbb{Z}_p \text{ for all } p, \quad \text{and} \quad \Lambda_1 \otimes \mathbb{R} \simeq \Lambda_2 \otimes \mathbb{R}.$$

Isometry of Lattices

Lattices Λ_1, Λ_2 are **isometric**, written $\Lambda_1 \simeq \Lambda_2$, if there is an isometry $f : V_1 \to V_2$ with $f(\Lambda_1) = \Lambda_2$.

Local-global for lattices. If $\Lambda_1 \simeq \Lambda_2$ as $\mathbb{Z}$-lattices, then

$$\Lambda_1 \otimes \mathbb{Z}_p \simeq \Lambda_2 \otimes \mathbb{Z}_p \text{ for all } p, \quad \text{and} \quad \Lambda_1 \otimes \mathbb{R} \simeq \Lambda_2 \otimes \mathbb{R}.$$

Unlike for quadratic forms over $\mathbb{Q}$, the converse is false for lattices over $\mathbb{Z}$!

Isometry of Lattices

Lattices Λ_1, Λ_2 are **isometric**, written $\Lambda_1 \simeq \Lambda_2$, if there is an isometry $f : V_1 \to V_2$ with $f(\Lambda_1) = \Lambda_2$.

Local-global for lattices. If $\Lambda_1 \simeq \Lambda_2$ as $\mathbb{Z}$-lattices, then

$$\Lambda_1 \otimes \mathbb{Z}_p \simeq \Lambda_2 \otimes \mathbb{Z}_p \text{ for all } p, \quad \text{and} \quad \Lambda_1 \otimes \mathbb{R} \simeq \Lambda_2 \otimes \mathbb{R}.$$

Unlike for quadratic forms over $\mathbb{Q}$, the converse is false for lattices over $\mathbb{Z}$!

Definition

Lattices with the same local data, $\Lambda_1 \otimes \mathbb{Z}_p \simeq \Lambda_2 \otimes \mathbb{Z}_p$ for all p, are said to lie in the same **genus**, written $\Lambda_1 \sim \Lambda_2$. The set of isometry classes within a genus is the **class set** $\mathrm{cls}(\Lambda)$.

Examples of Class Sets

Example 1.

$$\mathrm{cls}(E_{16}) = \{\, E_{16},\ E_8 \oplus E_8 \,\}.$$

Two non-isometric even unimodular lattices of rank 16.

Examples of Class Sets

Example 1.

$$\mathrm{cls}(E_{16}) = \{\, E_{16},\ E_8 \oplus E_8 \,\}.$$

Two non-isometric even unimodular lattices of rank 16.

Example 2. If

$$\Lambda_1 = \begin{pmatrix} 2 & 0 & 0 \\ 0 & 2 & 1 \\ 0 & 1 & 6 \end{pmatrix}, \qquad \Lambda_2 = \begin{pmatrix} 2 & -1 & 1 \\ -1 & 2 & -1 \\ 1 & -1 & 8 \end{pmatrix},$$

then

$$\mathrm{cls}(\Lambda_1) = \{\Lambda_1, \Lambda_2\}.$$

Examples of Class Sets

Example 1.

$$\mathrm{cls}(E_{16}) = \{\, E_{16},\ E_8 \oplus E_8 \,\}.$$

Two non-isometric even unimodular lattices of rank 16.

Example 2. If

$$\Lambda_1 = \begin{pmatrix} 2 & 0 & 0 \\ 0 & 2 & 1 \\ 0 & 1 & 6 \end{pmatrix}, \qquad \Lambda_2 = \begin{pmatrix} 2 & -1 & 1 \\ -1 & 2 & -1 \\ 1 & -1 & 8 \end{pmatrix},$$

then

$$\mathrm{cls}(\Lambda_1) = \{\Lambda_1, \Lambda_2\}.$$

Question. How do we detect isometries – and enumerate class sets – computationally? (We will answer this in two slides.)

Isometry of Lattices in Magma

```
> // Define lattices via Gram matrices (lower-triangular entries)
> L1 := LatticeWithGram(SymmetricMatrix([2, 0,2, 0,1,6]));
> // Recover the quadratic form (note the factor of 2 from B_Q = 2Q)
> QuadraticForm(L1) div 2;
x1^2 + x2^2 + x2*x3 + 3*x3^2
> L2 := LatticeWithGram(SymmetricMatrix([2,-1,2,1,-1,8]));
> QuadraticForm(L2) div 2;
x1^2 - x1*x2 + x1*x3 + x2^2 - x2*x3 + 4*x3^2
> // Check for isometry
> iso, T := IsIsometric(L1, L2);
> iso;
false
```

Computing the Genus: p-Neighbors

How do we enumerate $\mathbf{cls}(\Lambda)$? Kneser's idea:

Definition

Integral lattices $\Lambda, \Lambda' \subset V$ are **p-neighbors** if

$$[\Lambda : \Lambda \cap \Lambda'] = [\Lambda' : \Lambda \cap \Lambda'] = p.$$

Computing the Genus: p-Neighbors

How do we enumerate $\mathbf{cls}(\Lambda)$? Kneser's idea:

Definition

Integral lattices $\Lambda, \Lambda' \subset V$ are **p-neighbors** if

$$[\Lambda : \Lambda \cap \Lambda'] = [\Lambda' : \Lambda \cap \Lambda'] = p.$$

Theorem (Kneser)

If $p \nmid \mathrm{disc}(\Lambda)$, every lattice in the genus of Λ is reachable from Λ by a sequence of p-neighbor steps.

Computing the Genus: p-Neighbors

How do we enumerate $\mathbf{cls}(\Lambda)$? Kneser's idea:

Definition

Integral lattices $\Lambda, \Lambda' \subset V$ are **p-neighbors** if

$$[\Lambda : \Lambda \cap \Lambda'] = [\Lambda' : \Lambda \cap \Lambda'] = p.$$

Theorem (Kneser)

If $p \nmid \mathrm{disc}(\Lambda)$, every lattice in the genus of Λ is reachable from Λ by a sequence of p-neighbor steps.

Algorithm: BFS on the p-neighbor graph until no new isometry classes appear.

Computing the Genus: p-Neighbors

How do we enumerate $\mathbf{cls}(\Lambda)$? Kneser's idea:

Definition

Integral lattices $\Lambda, \Lambda' \subset V$ are **p-neighbors** if

$$[\Lambda : \Lambda \cap \Lambda'] = [\Lambda' : \Lambda \cap \Lambda'] = p.$$

Theorem (Kneser)

If $p \nmid \mathrm{disc}(\Lambda)$, every lattice in the genus of Λ is reachable from Λ by a sequence of p-neighbor steps.

Algorithm: BFS on the p-neighbor graph until no new isometry classes appear.

Bonus: counting neighbors yields the Hecke operator T_p — preview of Lecture 4.

p-Neighbors in Magma

```
> L := LatticeWithGram(SymmetricMatrix([2,0,2,0,1,6]));
> // 3-neighbors of L (3 does not divide disc(L) = 22)
> nbrs := Neighbors(L, 3);
> #nbrs;
2
> // Direct computation of the genus
> reps := GenusRepresentatives(L);
> #reps;
2
> // The two class representatives we saw earlier
> [* QuadraticForm(M) div 2 : M in reps *];
[*
    x1^2 + x2^2 + x2*x3 + 3*x3^2,
    x1^2 - x1*x2 + x1*x3 + x2^2 - x2*x3 + 4*x3^2
*]
```

How Many Genera? – A Research Vista

Question. How many genera of $\mathbb{Z}$-lattices of rank r and discriminant D are there?

Equivalently: how large can the LMFDB lattice tables grow?

Question. How many genera of $\mathbb{Z}$-lattices of rank r and discriminant D are there? Equivalently: how large can the LMFDB lattice tables grow?

Theorem (A., Chen, Garg, Wang 2026)

Let $\lambda = \frac{3+\sqrt{17}}{2}$. The number of genera of lattices of rank r and discriminant D is at most $\frac{1}{2} r \cdot D^{\log_2 \lambda + \pi(1+o(1))/\log\log D}$, where the $o(1)$ is as $D \to \infty$ with r fixed.

How Many Genera? – A Research Vista

Question. How many genera of $\mathbb{Z}$-lattices of rank r and discriminant D are there?
Equivalently: how large can the LMFDB lattice tables grow?

Theorem (A., Chen, Garg, Wang 2026)

Let $\lambda = \frac{3+\sqrt{17}}{2}$. The number of genera of lattices of rank r and discriminant D is at most $\frac{1}{2}r \cdot D^{\log_2 \lambda + \pi(1+o(1))/\log\log D}$, where the $o(1)$ is as $D \to \infty$ with r fixed.

What's tabulated today.

- Even unimodular lattices in low dimension: see Chenevier–Taïbi.
- LMFDB: `https://alpha.lmfdb.org/Lattice` – searchable by rank, discriminant, class number, theta series, …
- Enumeration uses Kneser's p-neighbors (just demonstrated).

Theta Functions

Beyond classification, lattices encode arithmetic in how often they represent integers. Define the **representation numbers**

$$r_\Lambda(m) = \#\{\lambda \in \Lambda : Q(\lambda) = m\}.$$

Example. For $\Lambda = \mathbb{Z}^2$, $Q = x^2 + y^2$: $r_\Lambda(1) = 4$, $r_\Lambda(2) = 4$, $r_\Lambda(5) = 8$, ...

Theta Functions

Beyond classification, lattices encode arithmetic in how often they represent integers. Define the **representation numbers**

$$r_\Lambda(m) = \#\{\lambda \in \Lambda : Q(\lambda) = m\}.$$

Example. For $\Lambda = \mathbb{Z}^2$, $Q = x^2 + y^2$: $r_\Lambda(1) = 4$, $r_\Lambda(2) = 4$, $r_\Lambda(5) = 8$, ...

Assume Q is positive definite. Package the $r_\Lambda(m)$ into a generating series:

Definition

The **theta function** of Λ is

$$\theta_\Lambda(q) \; = \sum_{\lambda \in \Lambda} q^{Q(\lambda)} \; = \sum_{m=0}^{\infty} r_\Lambda(m)\, q^m.$$

Theta Functions

Beyond classification, lattices encode arithmetic in how often they represent integers. Define the **representation numbers**

$$r_\Lambda(m) = \#\{\lambda \in \Lambda : Q(\lambda) = m\}.$$

Example. For $\Lambda = \mathbb{Z}^2$, $Q = x^2 + y^2$: $r_\Lambda(1) = 4$, $r_\Lambda(2) = 4$, $r_\Lambda(5) = 8$, ...

Assume Q is positive definite. Package the $r_\Lambda(m)$ into a generating series:

Definition

The **theta function** of Λ is

$$\theta_\Lambda(q) \;=\; \sum_{\lambda\in\Lambda} q^{Q(\lambda)} \;=\; \sum_{m=0}^{\infty} r_\Lambda(m)\, q^m.$$

Clearly $\Lambda_1 \simeq \Lambda_2 \implies \theta_{\Lambda_1} = \theta_{\Lambda_2}$. The converse is false – next slide.

Theta Series of E_8 and $E_8 \oplus E_8$

```
> E8 := Lattice("E", 8);
> theta_E8<q> := ThetaSeries(E8, 9);
> theta_E8;
1 + 240*q^2 + 2160*q^4 + 6720*q^6 + 17520*q^8 + O(q^10)
> // Direct sum: theta multiplies
> E8_2 := DirectSum(E8, E8);
> theta_E8_2 := ThetaSeries(E8_2, 9);
> theta_E8_2;
1 + 480*q^2 + 61920*q^4 + 1050240*q^6 + 7926240*q^8 + O(q^10)
> assert theta_E8^2 eq theta_E8_2;
```

$\theta_{E_{16}} = \theta_{E_8 \oplus E_8}$, but $E_{16} \not\cong E_8 \oplus E_8$

```
> D16 := Lattice("D", 16);
> E16 := D16 + Lattice(Matrix([[1/2 : i in [1..16]]]));
> theta_E16 := ThetaSeries(E16, 9);
> theta_E16;
1 + 480*q^2 + 61920*q^4 + 1050240*q^6 + 7926240*q^8 + O(q^10)
> assert theta_E16 eq theta_E8_2;
> // Same theta series, but...
> IsIsometric(E8_2, E16);
false
```

Conclusion: the theta function does not determine the lattice up to isometry ("one cannot hear the shape of a drum").

The LMFDB

The L-functions and Modular Forms Database

https://www.lmfdb.org

A searchable database of mathematical objects from number theory and arithmetic geometry, with extensive cross-references between them.

https://www.lmfdb.org

A searchable database of mathematical objects from number theory and arithmetic geometry, with extensive cross-references between them.

What's in it (relevant to us)?

- Number fields, elliptic curves, modular forms, L-functions, ...
- Integral lattices: *https://alpha.lmfdb.org/Lattice*
- Searchable by rank, discriminant, class num, minimum, theta series, name, ...

The L-functions and Modular Forms Database

https://www.lmfdb.org

A searchable database of mathematical objects from number theory and arithmetic geometry, with extensive cross-references between them.

What's in it (relevant to us)?

- Number fields, elliptic curves, modular forms, L-functions, ...
- Integral lattices: *https://alpha.lmfdb.org/Lattice*
- Searchable by rank, discriminant, class num, minimum, theta series, name, ...

Why use it?

- Quick lookups without re-running expensive computations.
- Authoritative reference data (verified, cross-checked).
- Cross-links: a lattice page connects to the associated genus, modular form, ...

Touring the Lattice Database

Introduction
Overview · Random
Universe · Knowledge
L-functions
Modular forms
Classical
Siegel
Varieties
Genus 2 curves over $\mathbb{Q}$
Shimura curves
Modular curves
Hilbert surfaces
Abelian varieties over $\mathbb{F}_q$
Fields
Representations
mod ℓ Galois
Motives
Groups

Refine search

Rank	Signature	Determinant	Discriminant	Level
3	[1,1]	10	11	48

Aut. group order	Aut. group	Class number	Minimum	Parity
2	8.1	5	1	even

Dual determinant	Dual kissing number	Disc. group invariants	Kissing number	Festi-Veniani Index
1	1	[2,4]	1	1

Gram matrix: auto — [5,1,23]

Search again · Random lattice · Diagram search · Sort order: ▲rank · Select: columns to display

Results (displaying both matches)

Download displayed columns for all results to Text

Label	Rank	Signature	Discriminant	Level	Class number	Parity	Determinant	Minimum	Aut. group order	Gram matrix
3.3.22.a6.1	3	$(3,0)$	11	44	2	Even	22	2	12	$\begin{pmatrix} 2 & 1 & -1 \\ 1 & 2 & -1 \\ -1 & -1 & 8 \end{pmatrix}$
3.3.22.a6.2	3	$(3,0)$	11	44	2	Even	22	2	8	$\begin{pmatrix} 2 & 0 & 0 \\ 0 & 2 & 1 \\ 0 & 1 & 6 \end{pmatrix}$

Touring the Lattice Database



Label	Rank	Signature	Discriminant	Level	Class number	Parity	Determinant	Minimum	Aut. group order	Gram matrix
3.3.22.a6.1	3	$(3,0)$	11	44	2	Even	22	2	12	$\begin{pmatrix} 2 & 1 & -1 \\ 1 & 2 & -1 \\ -1 & -1 & 8 \end{pmatrix}$
3.3.22.a6.2	3	$(3,0)$	11	44	2	Even	22	2	8	$\begin{pmatrix} 2 & 0 & 0 \\ 0 & 2 & 1 \\ 0 & 1 & 6 \end{pmatrix}$

Each lattice page records:

- Rank, determinant, level, class number, minimum, …
- A canonical Gram matrix.
- Theta series coefficients.
- Genus representatives (other lattices in $\mathrm{cls}(\Lambda)$).
- Names: E_8, D_n^+, Leech, …

Worked Example: Looking Up Λ_1

We just computed $\mathbf{cls}(\Lambda_1) = \{\Lambda_1, \Lambda_2\}$ in Magma in $\sim$ a second. Now let's verify – and discover more – via the LMFDB.

$$\Lambda_1 = \begin{pmatrix} 2 & 0 & 0 \\ 0 & 2 & 1 \\ 0 & 1 & 6 \end{pmatrix}, \qquad \mathbf{disc}(\Lambda_1) = 22.$$

Worked Example: Looking Up Λ_1

We just computed $\mathbf{cls}(\Lambda_1) = \{\Lambda_1, \Lambda_2\}$ in Magma in $\sim$ a second. Now let's verify – and discover more – via the LMFDB.

$$\Lambda_1 \;=\; \begin{pmatrix} 2 & 0 & 0 \\ 0 & 2 & 1 \\ 0 & 1 & 6 \end{pmatrix}, \qquad \mathbf{disc}(\Lambda_1) = 22.$$

Search: rank $= 3$, (half-)discriminant $= 11$ (current LMFDB), is_even.

Worked Example: Looking Up Λ_1

We just computed $\mathbf{cls}(\Lambda_1) = \{\Lambda_1, \Lambda_2\}$ in Magma in $\sim$ a second. Now let's verify – and discover more – via the LMFDB.

$$\Lambda_1 = \begin{pmatrix} 2 & 0 & 0 \\ 0 & 2 & 1 \\ 0 & 1 & 6 \end{pmatrix}, \qquad \mathrm{disc}(\Lambda_1) = 22.$$

Search: rank $= 3$, (half-)discriminant $= 11$ (current LMFDB), is_even.



Worked Example: Looking Up Λ_1

We just computed $\mathrm{cls}(\Lambda_1) = \{\Lambda_1, \Lambda_2\}$ in Magma in $\sim$ a second. Now let's verify – and discover more – via the LMFDB.

$$\Lambda_1 = \begin{pmatrix} 2 & 0 & 0 \\ 0 & 2 & 1 \\ 0 & 1 & 6 \end{pmatrix}, \qquad \mathrm{disc}(\Lambda_1) = 22.$$

Search: rank $= 3$, (half-)discriminant $= 11$ (current LMFDB), is_even.



Cross-References: the Web of LMFDB

Each LMFDB object links to related objects across the database.

From a lattice page, one can navigate to:

- the associated **quadratic form** (via Gram matrix);
- other members of the **genus**;
- attached **modular forms** (via theta series, when the rank is even);
- the underlying **number field** or **order** (for rank-2 forms).

Cross-References: the Web of LMFDB

Each LMFDB object links to related objects across the database.

From a lattice page, one can navigate to:

- the associated **quadratic form** (via Gram matrix);
- other members of the **genus**;
- attached **modular forms** (via theta series, when the rank is even);
- the underlying **number field** or **order** (for rank-2 forms).

Showcase. The theta series of E_8 is the Eisenstein series E_4, a weight-4 modular form on $\mathrm{SL}_2(\mathbb{Z})$. The LMFDB page for E_8 links to the page for E_4, which links to the trivial Dirichlet character, which links to the ζ-function, ...

Cross-References: the Web of LMFDB

Each LMFDB object links to related objects across the database.

From a lattice page, one can navigate to:

- the associated **quadratic form** (via Gram matrix);
- other members of the **genus**;
- attached **modular forms** (via theta series, when the rank is even);
- the underlying **number field** or **order** (for rank-2 forms).

Showcase. The theta series of E_8 is the Eisenstein series E_4, a weight-4 modular form on $\mathrm{SL}_2(\mathbb{Z})$. The LMFDB page for E_8 links to the page for E_4, which links to the trivial Dirichlet character, which links to the ζ-function, . . .

Theme of the mini-course: the LMFDB makes the web of connections in number theory *tangible* and *computable*. The next three lectures will traverse this web – number fields, modular forms, and back.

Summary

Key Takeaways

1. Magma is a powerful tool for computational algebra and number theory.
2. Quadratic forms encode rich number-theoretic information (representations, class numbers, …).
3. **Hasse–Minkowski:** equivalence of forms over $\mathbb{Q}$ is detected by finitely many local computations.
4. For lattices over $\mathbb{Z}$, the local-to-global principle *fails*; this gives rise to the genus and the class set.
5. Theta functions package representation numbers, but do *not* determine the lattice up to isometry.
6. The LMFDB provides a searchable database of these objects.

Next lecture: number fields and the Kronecker–Weber theorem.